
· In In the SAP Advanced Business Application Programming language (ABAP), you can use both Open SQL commands and corresponding database-specific Native SQL commands for performing database updates. 

· Accessing the database with Native SQL enables you to use database-specific commands. This requires detailed knowledge of the syntax in question. Programs that use Native SQL commands need additional programming after they have been transported to different system environments (different database systems), since the syntax of the SQL commands generally needs to be adjusted on a database-specific basis.

· Open SQL commands are not database-specific; they are automatically converted into the respective SQL statements by the database interface and passed to the database.  An ABAP program that operates with Open SQL is therefore not database-specific and you can use it in any R/3 System without having to adjust it.

· A further advantage of using Open SQL is that you can buffer SAP tables locally on the application server for quicker read access. This also means that the database load is reduced.  The data is read from the buffer automatically after the respective table settings have been made.

· The Open SQL set of commands includes operations for Data Manipulation Language (DML) only. Open SQL commands do not contain operations for Data Definition Language (DDL) because those operations are included in the ABAP Dictionary.

· You should implement database accesses using Native SQL only if a particular Native SQL function that is not available in Open SQL must be used. 

· For more information about Native SQL and Open SQL, refer to the ABAP Editor keyword documentation for SQL.

· You can limit the target quantity on the database using all the Open SQL commands discussed here.

· One or more rows can be processed using an SQL command. Commands that process several lines usually give better performance than corresponding single-set accesses (exception: mass data change using MODIFY).

· A syntax variant for the change operation allows you to change individual fields in a line.

· If you have masked field selections (WHERE <feld1> LIKE '<search_mask>'), note that '_' masks an individual character and '%' masks a character string of any length (in line with the SQL standard).

· All Open SQL commands provide you with a return message about the success or failure of the database operation performed. The message is issued in the form of a return code in the system field sy-subrc. Return code '0' (zero) always means that the operation has been completed successfully. All other values mean that errors have occurred. For further details, refer to the keyword documentation for the command in question.

· In addition, the sy-dbcnt system field displays the number of records for which the desired database operation was actually carried out.

· Open SQL commands do not perform any automatic authorization checks. You need to execute these explicitly in your program (refer to the topic in the appendix about authorization checks). 

· If the addition CLIENT SPECIFIED is not used  in an Open SQL command, no client specification is allowed in the respective WHERE clause, and the corresponding records of the current execution client are accessed. 

· If you wish to process data from other clients, you must specify the addition "CLIENT SPECIFIED" in the Open SQL command and the respective client(s) in the appropriate WHERE clause. 

· Make sure that an Open SQL command accesses all (!) clients if it contains the addition CLIENT SPECIFIED without a client specification.

· To insert a new row in a database table, enter the command INSERT INTO <dbtab> VALUES <wa>.  For this purpose, you must place the line to be inserted in the structure  <wa> before the command is called. This structure must be the same as for the lines in the database table concerned.

· The client field that might exist in the structure <wa> will be taken into consideration only if the CLIENT SPECIFIED addition is specified.   If there is no "CLIENT SPECIFIED" addition, the current execution client applies. 

· Rows can also be inserted using views. However, the view must already be created in the ABAP Dictionary with the maintenance status "read and change" and must only contain fields from a table.

· This INSERT variant has the following return codes:

· 0 = Line inserted successfully.

· 4 = Line could not be inserted because a line with the same key already exists.

· Alternative syntax :  INSERT <dbtab> [CLIENT SPECIFIED] FROM <wa>.

· You can use the command INSERT <dbtab> FROM TABLE <itab> to create several rows in a database table. The internal table <itab> that you should specify here must have the same line structure as the corresponding database table and the new data records. 

· The client field that might exist in the internal table <itab> will only be taken into consideration if the CLIENT SPECIFIED addition is specified.   If there is no CLIENT SPECIFIED addition, the current execution client applies. 

· If it is possible to create all the lines, sy-subrc is automatically set to zero.

· If, however, even one data record cannot be created, a runtime error occurs. This means that the entire insertion operation is discarded (database rollback). If, in such a case, you want to have records that can be inserted actually inserted, use the command addition ACCEPTING DUPLICATE KEYS. If there is an error, this addition suppresses  the runtime error (and thus also the database rollback), sets the return code (sy-subrc) to 4, and inserts all the records without errors.

· The sy-dbcnt system field contains the number of rows that were successfully inserted in the database.

· Using the two variants of the UPDATE command shown in the graphic, you can change a specific line within a database table.

· In variant 1, the database record containing the key in <wa> is overwritten by <wa>. However, the key field MANDT that might exist in <wa> is taken into consideration only if the CLIENT SPECIFIED addition is specified as well (otherwise the current execution client applies). Logically, <wa> must have the same structure as the database record to be changed.

· In variant 2, the record specified in the WHERE clause is changed. However, only the fields specified in the SET addition are overwritten on the database side with the values specified. In this syntax version, you must define the record to be changed in the WHERE clause by specifying all the key field evaluations. For details on specifying a possibly existing MANDT field, refer to the graphic, Accessing Client-Specific Tables, discussed previously in this unit. 

· You can specify simple calculation operations as evaulation for numeric database fields in the SET addition:  f = g ,   f = f + g ,   f = f - g . 

· Rows can also be changed using views. However, the view must already be created in the ABAP Dictionary with the maintenance status read and change and must only contain fields from a table.

· These two UPDATE variants have the following return codes :
 0 = Line was changed.
 4 = Line could not be changed because, for example, the specified key does not exist.

· If identical changes are to be made to the same fields in several rows of a database table, use the syntax specified on the slide.

· Using the WHERE clause you define which lines are to be changed. For details on specification of a possibly existing MANDT field, refer to the graphic, Accessing Client-Specific Tables, discussed previously in this unit. 

· In the SET addition you specify which fields in these records are to be changed.  The following calculations are also possible here for the numeric fields to be changed: 
 f = g ,   f = f + g ,   f = f - g . 

· This UPDATE variant has the following return codes:
 0 = At least one line has been changed.
 4 = No line was changed because, for example, no such line exists.

· The sy-dbcnt field contains the number of updated rows.

· You can perform a mass data change by specifying an internal table that has the same structure as the corresponding database table and the records to be changed.

· The MANDT field that possibly exists in the specified internal table is taken into consideration only if the CLIENT SPECIFIED addition is specified (otherwise the current execution client applies).

· This UPDATE variant has the following return codes:
 0 = All the specified lines were changed successfully.
 4 = At least one of the specified lines could not be changed (because, for example, it does not exist); 
      the other lines were changed.

· The sy-dbcnt system field contains the number of updated rows.

· The MODIFY command is SAP-specific.  It covers the two commands UPDATE and INSERT :

· If the data record specified in the MODIFY  statement exists, this record is updated (> UPDATE ).

· If the data record specified in the MODIFY  statement does not exist, this record is inserted (-> INSERT ).

· Using the various syntax variants, you can process single records and several records (same as the syntax of UPDATE and INSERT)

· The operation can also be carried out on views.  However, the view must already be created in the ABAP Dictionary with the maintenance status read and change and must contain only fields from a table.

· This command has the following return codes:
 0 = Specified record or all specified records were processed (updated/inserted).
 4 = The specified record or at least one of the specified records could not be processed - for example, because the record does not exist in the database and inserting it would destroy a unique secondary index. If you have chosen mass data change, the other records were processed.  

· The sy-dbcnt field contains the number of processed rows.

· The syntax specified in the graphic above for the DELETE command enables you to delete a single row in a database table.  In this syntax version, you must define the record to be changed in the WHERE clause by exactly specifying all the key field evaluations.  For details on specification of a possibly existing MANDT field, refer to the graphic, Accessing Client-Specific Tables.

· A row can also be deleted from views.  However, the view must already be created in the ABAP Dictionary with the maintenance status read and change and must contain only fields from a table.

· This DELETE variant has the following return codes:
 0 = Line was deleted.
 4 :=Line could not be deleted because, for example, it does not exist in the database.

· Alternative syntax :  DELETE <dbtab> [CLIENT SPECIFIED] FROM <wa>.
With this syntax version, the structure <wa> must have the same structure as the records in the respective database table and must be filled with the key fields of the records to be deleted before the command is called. However, the key field MANDT, which possibly exists in <wa>, is only taken into consideration if the CLIENT SPECIFIED addition is specified (otherwise the execution client applies). 

· This syntax variant of the DELETE command enables you to delete several lines in a database table.  Here you can specify the lines that are to be deleted in the WHERE clause.  For details on specification of a possibly existing MANDT field, refer to the graphic, Accessing Client-Specific Tables".

· You can delete all the lines in a cross-client database table using the following syntax. Here <field> is an arbitrary table field: 
  DELETE FROM <dbtab> WHERE <field> LIKE '%'.

· If you wish to delete all the lines of the execution client from a client-specific database table, you can use the same syntax :
 DELETE FROM <dbtab> WHERE <field> LIKE '%'
If you wish to delete all the existing lines from a client-specific database table, you must also specify the  CLIENT SPECIFIED addition in the command:
  DELETE FROM <dbtab> CLIENT SPECIFIED WHERE <field> LIKE '%'.
· Deleting data records using conditions produces the following return codes:
 0 = At least one line was deleted.
 4 = No line was deleted because, for example, the specified lines do not even exist.

· The system field sy-dbcnt contains the number of lines that have been deleted from the database.

.


· If you wish to delete several records from a database table, you can specify them first in an internal table that has the same structure as the respective database table and then use the above syntax of the DELETE command.  To do so, all you need to do is specify the key part of the records to be deleted in the internal table.   If the key field MANDT exists, then it will only be taken into consideration if the CLIENT SPECIFIED addition is specified (otherwise the current execution client applies).

· This DELETE variant has the following return codes:
 0 = All the lines specified in the internal table were deleted.
 4 = At least one line could not be deleted (for example, because it does not exist); the other records 
     were deleted.

· The number of lines deleted from the database is shown in the system field sy-dbcnt. 

· If an Open SQL statement that executes a change to the database returns a return code different than zero, you should make sure that the database is returned to the same status as before you attempted to make the respective change. You achieve this by performing a database rollback, which reverses all changes to the current database LUW (see next unit). 

· There are two ways of causing a database rollback: 
 - Sending a termination dialog message (MESSAGE A)
 - Using the ABAP statement ROLLBACK WORK
· The transmission of MESSAGE A causes a database rollback and terminates the program.   All other message types (E,W, I) also involve a dialog but do not trigger a database rollback.

· The ABAP statement ROLLBACK WORK, on the other hand, causes a database rollback without terminating the program.  In this case, you should be careful since the context has not been reset in the current program.

	Unit: Updating Databases With Open SQL

Topic:  Changing a Single Record

	At the conclusion of these exercises, you will be able to:

· Insert single records in database tables



	[image: image20.wmf]
	The program SAPBC414T_CREATE_CUSTOMER_01 enables new customer data to be entered in screen 100.

Extend this program to include the database dialog:  
After the function code SAVE is triggered (for example, by clicking the Save icon), the customer data is to be written to the database table SCUSTOM.

	Program:
SAPMZ##_CUSTOMER1
Transaction: 
Z##_CUSTOMER1
Template:
SAPBC414T_CREATE_CUSTOMER_01
Model solution:
SAPBC414S_CREATE_CUSTOMER_01


1-1
Copy the program template SAPBC414T_CREATE_CUSTOMER_01 with all subobjects to SAPMZ##_CUSTOMER1 (## is your group number). Assign transaction code Z##_CUSTOMER1 to the program.

1-2
The ABAP statements for the database dialog are encapsulated in the subroutine SAVE_SCUSTOM.  The subroutine has already been created (and is empty).

1-2-1
Insert the new customer data record in the database table SCUSTOM.  If the operation is successful, message S015 should be issued; if the operation is not successful, termination message A048 should be issued.

	[image: image22.wmf]
	There are SCUSTOM fields on the input screen.

On the internal program side, the newly entered customer data is retained in the SCUSTOM structure defined in the TABLES-statement.

The call already implemented from the NUMBER_GET_NEXT subroutine places a new customer number into the field SCUSTOM-ID so that all the data of the new customer is completely available in the SCUSTOM structure.

The BC414 message class is declared as a program default in the PROGRAM statement and therefore valid throughout the programs.

	Unit: Updating Databases with Open SQL

Topic:  Changing Several Data Records (Optional)

	At the conclusion of these exercises, you will be able to:

· Change several data records, with respect to performance, in database tables.



	[image: image25.wmf]
	In the program SAPBC414T_UPDATE_STRAVELAG, a list is generated that presents the data of the travel agencies maintained in the STRAVELAG table.  By selecting one or more lines, the user can change the data of the corresponding travel agency on the subsequent screen (100). 

Extend the program to include the database dialog.
The changed data is to be saved to the STRAVELAG database table by clicking the Save icon (function code SAVE) on screen 100.

	Program:

SAPMZ##_UPDATE_STRAVELAG
Template:
SAPBC414T_UPDATE_STRAVELAG
Model solution:
SAPBC414S_UPDATE_STRAVELAG


2-1
Copy the program template SAPBC414T_UPDATE_STRAVELAG with all subobjects to SAPMZ##_UPDATE_STRAVELAG (## is the group number).  As this is a type 1 program, a transaction code is not required. 

2-2
The database dialog is initiated by triggering the function code SAVE.  Here, the SAVE_CHANGES subroutine, which contains the database dialog, is called in the PAI module USER_COMMAND_0100 (screen 100).   This subroutine has already been created (empty).

2-2-1
Save the changed address data to the database table STRAVELAG.  When doing so, note the performance aspects.  If the save is successful, the system issues the message S030; if it is not successful, the system issues the information message I048.

	The travel agency data is buffered in the internal table ITAB_TRAVEL (work area WA_TRAVEL). The line structure in the internal table has the same structure as that in STRAVELAG, with the exception of the additional field MARK_CHANGED (C(1)).   If the address data on the screen 100 has been changed, MARK_CHANGED has the value 'X'. Otherwise it is blank or 0.


0.2 Tips on Model Solutions for this Course

	The model solutions provided here repeat the statements of the flow logic and ABAP program parts that will be required.

The exercises for course BC414 are designed to expand on two larger programs accompanying the contents of the unit in question. For the sake of clarity, not all of the model solutions are provided with complete coding.  The following procedure is used instead:

· The model solution for the exercise in which a program is edited for the first time is displayed completely.

· Any model solutions that expand on this only explain flow logic, subroutines, and modules that have changed or appear for the first time. Boldface is used to highlight the statements in the repeated modularization units that need to be entered to solve the exercise.

· A complete version of both programs is provided in the appendix.

The second exercise in this unit, Updating Databases with Open SQL, which is marked as optional, is an exception to this procedure. Since the program associated with this activity is not dealt with in the following units, the model solution for this activity is explained fully.

Solutions

	Unit: Updating Databases with Open SQL

Topic: Changing a Single Record


Model Solution SAPBC414S_CREATE_CUSTOMER_01

2 Module Pool

*&-----------------------------------------------------------------*

*& Modulpool         SAPBC414S_CREATE_CUSTOMER_01                  *
*&-----------------------------------------------------------------*

INCLUDE BC414S_CREATE_CUSTOMERTOP.
INCLUDE BC414S_CREATE_CUSTOMERO01.
INCLUDE BC414S_CREATE_CUSTOMERI01.
INCLUDE BC414S_CREATE_CUSTOMER_01F01.
3 SCREEN 100
PROCESS BEFORE OUTPUT.
  MODULE status_0100.
PROCESS AFTER INPUT.
  MODULE exit AT EXIT-COMMAND.
  MODULE save_ok_code.
  FIELD: scustom-name MODULE mark_changed ON REQUEST.
  MODULE user_command_0100.
4 TOP Include
*&-----------------------------------------------------------------*

*& Include BC414S_CREATE_CUSTOMERTOP                               *
*&-----------------------------------------------------------------*

PROGRAM  sapbc414s_create_customer MESSAGE-ID bc414.
DATA: answer, flag.
DATA: ok_code LIKE sy-ucomm, save_ok LIKE ok_code.
TABLES: scustom.
5 PBO Module
*------------------------------------------------------------------*

***INCLUDE BC414S_CREATE_CUSTOMERO01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Module  STATUS_0100  OUTPUT
*&-----------------------------------------------------------------*

MODULE STATUS_0100 OUTPUT.
  SET PF-STATUS 'DYN_0100'.
  SET TITLEBAR 'DYN_0100'.
ENDMODULE.                 " STATUS_0100  OUTPUT
6 PAI Module
*------------------------------------------------------------------*

***INCLUDE BC414S_CREATE_CUSTOMERI01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Module  EXIT  INPUT
*&-----------------------------------------------------------------*

MODULE exit INPUT.
  CASE ok_code.
    WHEN 'EXIT'.
      IF sy-datar IS INITIAL AND flag IS INITIAL.
* no changes on screen 100
        LEAVE PROGRAM.
      ELSE.
        PERFORM ask_save USING answer.
        CASE answer.
          WHEN 'J'.
            ok_code = 'SAVE&EXIT'.
          WHEN 'N'.
            LEAVE PROGRAM.
          WHEN 'A'.
            CLEAR ok_code.
            SET SCREEN 100.
        ENDCASE.
      ENDIF.
    WHEN 'CANCEL'.
      IF sy-datar IS INITIAL AND flag IS INITIAL.
* no changes on screen 100
        LEAVE TO SCREEN 0.
      ELSE.
        PERFORM ask_loss USING answer.
        CASE answer.
          WHEN 'J'.
            LEAVE TO SCREEN 0.
          WHEN 'N'.
            CLEAR ok_code.
            SET SCREEN 100.
        ENDCASE.
      ENDIF.
  ENDCASE.
ENDMODULE.                             " EXIT  INPUT
*&-----------------------------------------------------------------*

*&      Module  SAVE_OK_CODE  INPUT
*&-----------------------------------------------------------------*

MODULE save_ok_code INPUT.
  save_ok = ok_code.
  CLEAR ok_code.
ENDMODULE.                             " SAVE_OK_CODE  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0100 INPUT.
  CASE save_ok.
    WHEN 'SAVE&EXIT'.
      PERFORM save.
      LEAVE PROGRAM.
    WHEN 'SAVE'.
      IF flag IS INITIAL.
        SET SCREEN 100.
      ELSE.
        PERFORM save.
        SET SCREEN 0.
      ENDIF.
    WHEN 'BACK'.
      IF flag IS INITIAL.
        SET SCREEN 0.
      ELSE.
        PERFORM ask_save USING answer.
        CASE answer.
          WHEN 'J'.
            PERFORM save.
            SET SCREEN 0.
          WHEN 'N'.
            SET SCREEN 0.
          WHEN 'A'.
            SET SCREEN 100.
        ENDCASE.
      ENDIF.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

*&      Module  MARK_CHANGED  INPUT
*&-----------------------------------------------------------------*

MODULE mark_changed INPUT.
* set flag to mark changes were made on screen 100
  flag = 'X'.
ENDMODULE.                             " MARK_CHANGED  INPUT
7 FORM Routines

*------------------------------------------------------------------*

***INCLUDE BC414S_CREATE_CUSTOMER_01F01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  NUMBER_GET_NEXT
*&-----------------------------------------------------------------*

*      -->P_WA_SCUSTOM  text
*------------------------------------------------------------------*

FORM number_get_next USING p_scustom LIKE scustom.
  DATA: return TYPE inri-returncode.
* get next free number in the number range '01'
* of number range object 'SBUSPID'
  CALL FUNCTION 'NUMBER_GET_NEXT'
       EXPORTING
            nr_range_nr = '01'
            object      = 'SBUSPID'
       IMPORTING
            number      = p_scustom-id
            returncode  = return
       EXCEPTIONS
            OTHERS      = 1.
  CASE sy-subrc.
    WHEN 0.
      CASE return.
        WHEN 1.
* number of remaining numbers critical
          MESSAGE s070.
        WHEN 2.
* last number
          MESSAGE s071.
        WHEN 3.
* no free number left over
          MESSAGE a072.
      ENDCASE.
    WHEN 1.
* internal error
      MESSAGE a073 WITH sy-subrc.
  ENDCASE.
ENDFORM.                               " NUMBER_GET_NEXT
*&-----------------------------------------------------------------*

*&      Form  ASK_SAVE
*&-----------------------------------------------------------------*

*      -->P_ANSWER  text
*------------------------------------------------------------------*

FORM ask_save USING p_answer.
  CALL FUNCTION 'POPUP_TO_CONFIRM_STEP'
       EXPORTING
            textline1 = 'Data has been changed.'(001)
            textline2 = 'Save before leaving transaction?'(002)
            titel     = 'Create Customer'(003)
       IMPORTING
            answer    = p_answer.
ENDFORM.                               " ASK_SAVE
*&-----------------------------------------------------------------*

*&      Form  ASK_LOSS
*&-----------------------------------------------------------------*

*      -->P_ANSWER  text
*------------------------------------------------------------------*

FORM ask_loss USING p_answer.
  CALL FUNCTION 'POPUP_TO_CONFIRM_LOSS_OF_DATA'
       EXPORTING
            textline1 = 'Continue?'(004)
            titel     = 'Create Customer'(003)
       IMPORTING
            answer    = p_answer.
ENDFORM.                               " ASK_LOSS
*&-----------------------------------------------------------------*

*&      Form  ENQ_SCUSTOM
*&-----------------------------------------------------------------*

FORM enq_scustom.
  CALL FUNCTION 'ENQUEUE_ESCUSTOM'
       EXPORTING
            id             = scustom-id
       EXCEPTIONS
            foreign_lock   = 1
            system_failure = 2
            OTHERS         = 3.
  CASE sy-subrc.
    WHEN 0.
    WHEN 1.
      MESSAGE e060.
    WHEN OTHERS.
      MESSAGE e063 WITH sy-subrc.
  ENDCASE.
ENDFORM.                               " ENQ_SCUSTOM
*&-----------------------------------------------------------------*

*&      Form  DEQ_ALL
*&-----------------------------------------------------------------*

FORM deq_all.
  CALL FUNCTION 'DEQUEUE_ALL'.
ENDFORM.                               " DEQ_ALL
*&-----------------------------------------------------------------*

*&      Form  SAVE
*&-----------------------------------------------------------------*

FORM save.
* get SCUSTOM-ID from number range object SBUSPID
  PERFORM number_get_next USING scustom.
* save new customer
  PERFORM save_scustom.
ENDFORM.                               " SAVE
*&-----------------------------------------------------------------*

*&      Form  SAVE_SCUSTOM
*&-----------------------------------------------------------------*

FORM save_scustom.
  INSERT INTO scustom VALUES scustom.
  IF sy-subrc <> 0.
* insertion of dataset in DB-table not possible
    MESSAGE a048.
  ELSE.
* insertion successfull
    MESSAGE s015 WITH scustom-id.
  ENDIF.
ENDFORM.                               " SAVE_SCUSTOM
Solutions

	Unit: Updating Databases With Open SQL

Topic: Changing Several Data Records


Model Solution SAPBC414S_UPDATE_STRAVELAG

8 Module Pool

*&-----------------------------------------------------------------*

*& Module Pool         SAPBC414S_UPDATE_STRAVELAG                    *

*&-----------------------------------------------------------------*

INCLUDE bc414s_update_stravelagtop.
INCLUDE bc414s_update_stravelagf01.
INCLUDE bc414s_update_stravelago01.
INCLUDE bc414s_update_stravelagi01.
INCLUDE bc414s_update_stravelage01.
9 SCREEN 100
PROCESS BEFORE OUTPUT.
  MODULE STATUS_0100.
* fill table control (only agencies, marked on list)
  LOOP AT ITAB_TRAVEL INTO WA_TRAVEL WITH CONTROL TC_STRAVELAG.
    MODULE TRANS_TO_DYNPRO.
  ENDLOOP.
*

PROCESS AFTER INPUT.
  MODULE EXIT AT EXIT-COMMAND.
  LOOP AT ITAB_TRAVEL.
    CHAIN.
     FIELD: STRAVELAG-STREET, STRAVELAG-POSTBOX, STRAVELAG-POSTCODE,
            STRAVELAG-CITY, STRAVELAG-COUNTRY, STRAVELAG-REGION,
            STRAVELAG-TELEPHONE, STRAVELAG-URL, STRAVELAG-LANGU.
* mark datasets, that were changed in table control (subset of all
* agencies, thet were shown on table control)
      MODULE SET_MARKER ON CHAIN-REQUEST.
    ENDCHAIN.
  ENDLOOP.
  MODULE SAVE_OK_CODE.
  MODULE USER_COMMAND_0100.
TOP Include
*&-----------------------------------------------------------------*

*& Include BC414S_UPDATE_STRAVELAGTOP                              *
*&-----------------------------------------------------------------*

PROGRAM  sapbc414s_update_stravelag NO STANDARD PAGE HEADING
                                    LINE-SIZE 120
                                    LINE-COUNT 10
                                    MESSAGE-ID bc414.
* Line type definition for internal table itab_travel
TYPES: BEGIN OF stravel_type.
        INCLUDE STRUCTURE stravelag.
TYPES:            mark_changed,
       END OF stravel_type.
* Standard internal table for travel agency data buffering and
* corresponding workarea
DATA: itab_stravelag LIKE STANDARD TABLE OF stravelag
      WITH NON-UNIQUE KEY agencynum,
      wa_stravelag TYPE stravelag.
* Workarea for transport of field values from/to screen 100
TABLES: stravelag.
* Transport function code from screen 100
DATA: ok_code TYPE sy-ucomm, save_ok LIKE ok_code.
* Table control structure on screen 100
CONTROLS: tc_stravelag TYPE TABLEVIEW USING SCREEN '0100'.
* Internal table to collect marked list entries, corresponding 
* workarea
DATA: itab_travel TYPE STANDARD TABLE OF stravel_type
      WITH NON-UNIQUE KEY agencynum,
      wa_travel TYPE stravel_type.
* Mark field displayed as checkbox on list
DATA: mark.
* Flags:
DATA: flag,         "changes performed on table control
      modify_list.  "modification of list buffer is neccessary
* Positions of fields on list
CONSTANTS: pos1 TYPE i VALUE   1,
           pos2 TYPE i VALUE   3,
           pos3 TYPE i VALUE  14,
           pos4 TYPE i VALUE  40,
           pos5 TYPE i VALUE  71,
           pos6 TYPE i VALUE  82,
           pos7 TYPE i VALUE 108.
PBO Modules

*------------------------------------------------------------------*

***INCLUDE BC414S_UPDATE_STRAVELAGO01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Module  STATUS_0100  OUTPUT
*&-----------------------------------------------------------------*

MODULE status_0100 OUTPUT.
  SET PF-STATUS 'DYNPRO'.
  SET TITLEBAR 'DYNPRO'.
ENDMODULE.                             " STATUS_0100  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  TRANS_TO_DYNPRO  OUTPUT
*&-----------------------------------------------------------------*

MODULE trans_to_dynpro OUTPUT.
* Field transport to screen
  MOVE-CORRESPONDING wa_travel TO stravelag.
ENDMODULE.                             " TRANS_TO_DYNPRO  OUTPUT
9.2 PAI Modules

*------------------------------------------------------------------*

***INCLUDE BC414S_UPDATE_STRAVELAGI01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0100 INPUT.
  CASE save_ok.
    WHEN 'SAVE'.
      IF flag IS INITIAL.
* enries on table control not changed.
        SET SCREEN 0.
      ELSE.
* at least one field on table control changed
        PERFORM save_changes.
        SET SCREEN 0.
      ENDIF.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

*&      Module  SAVE_OK_CODE  INPUT
*&-----------------------------------------------------------------*

MODULE save_ok_code INPUT.
  save_ok = ok_code.
  CLEAR: ok_code.
ENDMODULE.                             " SAVE_OK_CODE  INPUT
*&-----------------------------------------------------------------*

*&      Module  EXIT  INPUT
*&-----------------------------------------------------------------*

MODULE exit INPUT.
  CASE ok_code.
    WHEN 'CANCEL'.
      IF sy-datar IS INITIAL AND flag IS INITIAL.
* no changes performed on screen
        LEAVE TO SCREEN 0.
      ELSE.
* at least one field on table control changed.
        PERFORM popup_to_confirm_loss_of_data.
      ENDIF.
  ENDCASE.
ENDMODULE.                             " EXIT  INPUT
*&-----------------------------------------------------------------*

*&      Module  SET_MARKER  INPUT
*&-----------------------------------------------------------------*

MODULE set_marker INPUT.
  MOVE-CORRESPONDING stravelag TO wa_travel.
  wa_travel-mark_changed = 'X'.
* mark datasets in internal table as modified
  MODIFY TABLE itab_travel FROM wa_travel.
* at least one dataset is modified in table control
  flag = 'X'.
ENDMODULE.                             " SET_MARKER  INPUT
Events

*------------------------------------------------------------------*

*   INCLUDE BC414S_UPDATE_STRAVELAGE01                             *
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&   Event START-OF-SELECTION
*&-----------------------------------------------------------------*

START-OF-SELECTION.
* Read data from STRAVELAG into internal table ITAB_STRAVELAG
  PERFORM read_data USING itab_stravelag.
* Write data from ITAB_STRAVELAG on list
  PERFORM write_data.
*&-----------------------------------------------------------------*

*&   Event TOP-OF-PAGE
*&-----------------------------------------------------------------*

TOP-OF-PAGE.
* Write page title and page heading
  PERFORM write_header.
*&-----------------------------------------------------------------*

*&   Event END-OF-SELECTION
*&-----------------------------------------------------------------*

END-OF-SELECTION.
* Set PF-Status and Title of list
  SET PF-STATUS 'LIST'.
  SET TITLEBAR 'LIST'.
*&-----------------------------------------------------------------*

*&   Event AT USER-COMMAND
*&-----------------------------------------------------------------*

AT USER-COMMAND.
  CLEAR: modify_list, flag, itab_travel.
* Collect data corresponding to marked lines into internal table
  PERFORM loop_at_list USING itab_travel.
* Call screen if any line on list was marked
  CHECK NOT itab_travel IS INITIAL.
  PERFORM call_screen.
* Modify list buffer if database table was modified -> submit report
  CHECK NOT modify_list IS INITIAL.
  SUBMIT (sy-cprog).
FORM Routines

*------------------------------------------------------------------*

***INCLUDE BC414S_UPDATE_STRAVELAGF01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  READ_DATA
*&-----------------------------------------------------------------*

*      -->P_ITAB_STRAVELAG  text
*------------------------------------------------------------------*

FORM read_data USING p_itab_stravelag LIKE itab_stravelag.
  SELECT * FROM stravelag
           INTO CORRESPONDING FIELDS OF TABLE p_itab_stravelag.
ENDFORM.                               " READ_DATA
*&-----------------------------------------------------------------*

*&      Form  WRITE_DATA
*&-----------------------------------------------------------------*

FORM write_data.
  LOOP AT itab_stravelag INTO wa_stravelag.
    WRITE AT: /pos1 mark AS CHECKBOX,
               pos2 wa_stravelag-agencynum COLOR COL_KEY,
               pos3 wa_stravelag-name,
               pos4 wa_stravelag-street,
               pos5 wa_stravelag-postcode,
               pos6 wa_stravelag-city,
               pos7 wa_stravelag-country.
    HIDE: wa_stravelag.
  ENDLOOP.
ENDFORM.                               " WRITE_DATA
*&-----------------------------------------------------------------*

*&      Form  WRITE_HEADER
*&-----------------------------------------------------------------*

FORM write_header.
  WRITE: / 'Travel agency data'(007), AT sy-linsz sy-pagno.
  ULINE.
  FORMAT COLOR COL_HEADING.
  WRITE AT: /pos2 'Agency'(001),
             pos3 'Name'(002),
             pos4 'Street'(003),
             pos5 'Postal Code'(004),
             pos6 'City'(005),
             pos7 'Country'(006).
  ULINE.
ENDFORM.                               " WRITE_HEADER
*&-----------------------------------------------------------------*

*&      Form  LOOP_AT_LIST
*&-----------------------------------------------------------------*

*      -->P_ITAB_AGNECYNUM  text
*------------------------------------------------------------------*

FORM loop_at_list USING p_itab_travel LIKE itab_travel.
  DO.
    CLEAR: mark.
    READ LINE sy-index FIELD VALUE mark.
    IF sy-subrc <> 0.
      EXIT.
    ENDIF.
    CHECK NOT mark IS INITIAL.
    APPEND wa_stravelag TO p_itab_travel.
  ENDDO.
ENDFORM.                               " LOOP_AT_LIST
*&-----------------------------------------------------------------*

*&      Form  CALL_SCREEN
*&-----------------------------------------------------------------*

FORM call_screen.
* Initialize table control on screen
  REFRESH CONTROL 'TC_STRAVELAG' FROM SCREEN '0100'.
* Show screen in modal dialog box.
  CALL SCREEN 100 STARTING AT  5  5
                  ENDING   AT 80 15.
ENDFORM.                               " CALL_SCREEN
*&-----------------------------------------------------------------*

*&      Form  POPUP_TO_CONFIRM_LOSS_OF_DATA
*&-----------------------------------------------------------------*

FORM popup_to_confirm_loss_of_data.
  DATA answer.
  CALL FUNCTION 'POPUP_TO_CONFIRM_LOSS_OF_DATA'
       EXPORTING
            textline1 = 'Cancel processing of travel agencies?'(008)
            titel     = 'Cancel processing'(009)
       IMPORTING
            answer    = answer.
  CASE answer.
    WHEN 'J'.
      LEAVE TO SCREEN 0.
    WHEN 'N'.
      LEAVE TO SCREEN '0100'.
  ENDCASE.
ENDFORM.                               " POPUP_TO_CONFIRM_LOSS_OF_DATA
*&-----------------------------------------------------------------*

*&      Form  SAVE_CHANGES
*&-----------------------------------------------------------------*

FORM save_changes.
* declare internal table and workarea of same linetype as DB table
  DATA: itab TYPE STANDARD TABLE OF stravelag,
        wa LIKE LINE OF itab.
* search for datasets changed on the screen
  LOOP AT itab_travel INTO wa_travel
       WHERE mark_changed = 'X'.
* fill workarea fitting to DB table
    MOVE-CORRESPONDING wa_travel TO wa.
* fill corresponding internal table
    APPEND wa TO itab.
  ENDLOOP.
* mass update on stravelag -> best performance
  UPDATE stravelag FROM TABLE itab.
* check success
  IF sy-subrc = 0.
* all datasets are successfully updated
    MESSAGE s030.
  ELSE.
* at least one dataset from the internal table could not be updated 
* on the database table
    MESSAGE i048.
  ENDIF.
* Flag: List does not show correct data any more
  modify_list = 'X'.
ENDFORM.                               " SAVE_CHANGES

